Solution to Assignment 2

Question 1

A possible solution is shown below.

```
-- ELEC 379 Assignment 2 Solution
-- MC74AC323 bidirectional shift/store 8-bit register
-- Ed Casas, Jan 25 1998

library ieee ;
use ieee.std_logic_1164.all ;

entity MC74AC323 is
   port (
   io : inout std_logic_vector (0 to 7) ;
   s0, s1, oe1, oe2, ds0, ds7, cp, sr : in std_logic ;
   q0, q7 : out std_logic ) ;
end  MC74AC323 ;


architecture rtl of MC74AC323 is
   signal q : std_logic_vector (0 to 7) ;
   signal s : std_logic_vector (1 downto 0) ;
begin

   -- tri-state control
   io <=
      q when ( ( oe1 = '0' ) and ( oe2 = '0' ) and
         ( s /= "11" ) ) else
      "ZZZZZZZZ" ;

   -- LS and MS bit outputs
   q0 <= q(0) ;
   q7 <= q(7) ;

   -- shift control
   s <= s1 & s0 ;

   process(cp,sr,s)
   begin
      if cp'event and cp = '1' then
         if sr = '0' then
            q <= "00000000" ;
         else
    case s is
            when "11" => q <= io ;
            when "01" => q <= ds0 & q(0 to 6) ;
            when "10" => q <= q(1 to 7) & ds7 ;
            when others => q <= q ;
            end case ;
         end if ;
      end if ;
   end process ;

end rtl ;
```

The resulting schematic is: